Índice general

1. Introducción

En este tema nos vamos a centrar en el estudio de los conjuntos ordenados, y de algunos tipos especiales de órdenes. Como tipos especiales de conjuntos ordenados destacaremos los retículos y álgebras de Boole. Daremos además caracterizaciones de éstos en términos de un conjunto con dos operaciones que cumplen una serie de propiedades, ofreciendo así una definición alternativa que es muy utilizada en la literatura, y que no hace uso del concepto de conjunto ordenado.

Finalizaremos este capítulo con algunas aplicaciones al diseño y simplificación de circuitos lógicos.

2. Conjuntos ordenados.

Antes de comenzar, vamos a definir en Mathematica algunas funciones básicas sobre conjuntos que necesitaremos en el resto del capítulo.

Una función para calcular el conjunto potencia de un conjunto.

```
subconjuntos[{}]:={{}};
subconjuntos[tor:{x_,xs___}]:=With[{rec=subconjuntos[{xs}]},
   Union[rec,Map[Join[{x},#]&,rec]]
 ]
```

```
subconjuntos[{1,2,3}]
```

{{}, {1}, {2}, {3}, {1,2}, {1,3}, {2,3}, {1,2,3}}

La función contenido

```
{}⊆_:=True;
{x_,xs___}⊆b_:={xs}⊆b /; MemberQ[b,x]
_⊆_:=False;
```

```
{1,2}⊆{1,2,3}
```

True

```
{2,1,3}⊆{1,2,3,4,5}
```

True

```
{2,1,3}⊆{3,4,1}
```

False

El producto cartesiano

```
cartesiano[{},y_]:={};
cartesiano[{x_,xs___},y_]:=Union[cartesiano[{xs},y], Map[{x,#}&,y]]
```

```
cartesiano[x_]:=cartesiano[x,x]
```

```
cartesiano[{1,2,3}]
```

{{1,1}, {1,2}, {1,3}, {2,1}, {2,2}, {2,3}, {3,1}, {3,2}, {3,3}}

Definición 1. Sea X un conjunto, y $\leq$ una relación binaria en X. Se dice que $\leq$ es una relación de orden si se verifican las siguientes propiedades.

Reflexiva: $x \leq x$ para todo $x \in X$.

Antisimétrica: Si $x \leq y$ e $y \leq x$ entonces $x = y$.

Transitiva: Si $x \leq y$ e $y \leq z$ entonces $x \leq z$.

Si X es un conjunto en el que tenemos definida una relación de orden $\leq$, se dice que $(X, \leq)$ es un conjunto ordenado (o, si está claro cual es la relación $\leq$ se dice simplemente que X es un conjunto ordenado).

Si $\leq$ es una relación de orden en X que satisface la propiedad adicional de que dados $x, y \in X$ entonces $x \leq y$ ó $y \leq x$, se dice entonces que $\leq$ es una relación de orden total, y que $(X, \leq)$ (o X) es un conjunto totalmente ordenado (en ocasiones, para destacar que $(X, \leq)$ es una relación de orden, pero que no es total se dice que $\leq$ es una relación de orden parcial y que $(X, \leq)$ es un conjunto parcialmente ordenado).

Ejemplo 2.

1. El conjunto de los números naturales, con el orden natural ($m \leq n$ si existe $k \in \mathbb{N}$ tal que $n = m + k$) es un conjunto totalmente ordenado. De la misma forma, también lo son $(\mathbb{Z}, \leq)$, $(\mathbb{Q}, \leq)$ y $(\mathbb{R}, \leq)$.

2. Dado un conjunto X, entonces $\mathscr{P}(X)$, con el orden dado por la inclusión es un conjunto ordenado. Si X tiene más de un elemento, este orden no es total, pues dados $x, y \in X$ distintos se tiene que $\{x\} \not\subseteq \{y\}$ y $\{y\} \not\subseteq \{x\}$.

3. En el conjunto de los números naturales, la relación de divisibilidad es una relación de orden que no es total. Sin embargo, en el conjunto de los números enteros esta relación no es de orden pues no es antisimétrica, ya que $2|-2$, $-2|2$ y sin embargo $2 \neq -2$.

4. Para cualquier número natural n consideramos el conjunto

$$\mathrm{D}(n) = \{m \in \mathbb{N} : m|n\}$$

   Entonces $(\mathrm{D}(n), |)$ es un conjunto (parcialmente) ordenado.

5. Sea $(X, \leq)$ es un conjunto ordenado, e Y un subconjunto de X. Definimos en Y el orden $x \preceq y$ si $x \leq y$ (vistos como elementos de X). Entonces, $(Y, \preceq)$ es un conjunto ordenado. De ahora en adelante, el orden en Y lo denotaremos igual que en X.

   Si $(X, \leq)$ es un conjunto totalmente ordenado, entonces, para cualquier $Y \subseteq X$ se tiene que $(Y, \leq)$ es un conjunto totalmente ordenado.

Dado un conjunto X y una relación R binaria en ese conjunto, representaremos la relación como un subconjunto de X×X.

Reflexiva

```
reflexivaQ[{{as___,x_,bs___},{es___,{x_,x_},fs___}}]:= reflexivaQ[{{as,bs},{es,fs}}];
reflexivaQ[{{},_}]:=True;
reflexivaQ[{_,_}]:=False;
```

```
reflexivaQ[{{1,2}, {{1,1}, {1,2}}}]
```

False

```
reflexivaQ[{{1,2}, {{1,1}, {1,2}, {2,2}}}]
```

True

Antisimétrica

```
antisimetricaQ[{{as___,x_,bs___,y_,cs___},{es___,{x_,y_},fs___,{y_,x_}, gs___}}]:= x===y;
antisimetricaQ[{{as___,y_,bs___,x_,cs___},{es___,{x_,y_},fs___,{y_,x_}, gs___}}]:= x===y;
antisimetricaQ[{_,_}]:=True;
```

```
antisimetricaQ[{{1,2}, {{1,1}, {1,2}, {2,2}}}]
```

True

```
antisimetricaQ[{{1,2}, {{2,1}, {1,2}, {1,1}}}]
```

False

Transitiva

```
transitivaQ[{v_,rel:{es___,{x_,y_},fs___,{y_,z_},gs___}}]:= False/;Not[MemberQ[rel,{x,z}]];
transitivaQ[{v_,rel:{es___,{y_,z_},fs___,{x_,y_},gs___}}]:= False/;Not[MemberQ[rel,{x,z}]];
transitivaQ[{_,_}]:=True;
```

Órdenes

```
ordenQ[q:{v_,r_}]:=reflexivaQ[q]∧antisimetricaQ[q]∧transitivaQ[q]

ordentotalQ[{{as___,x_,bs___,y,cs___},rel_}]:=
   False /; Not[MemberQ[rel, {x,y}]∨MemberQ[rel, {y,x}]]
ordentotalQ[q_]:=ordenQ[q];

relconj:=Select[cartesiano[x], (#[[1]]⊆#[[2]])&]

x=subconjuntos[{1,2,3}]
```

{{}, {1}, {2}, {3}, {1,2}, {1,3}, {2,3}, {1,2,3}}

```
cartesiano[x]
```

{{{}, {}}, {{}, {1}}, {{}, {2}}, {{}, {3}}, {{}, {1,2}}, {{}, {1,3}}, {{}, {2,3}},{{}, {1,2,3}}, {{1}, {}}, {{1}, {1}}, {{1}, {2}}, {{1}, {3}}, {{1}, {1,2}}, {{1}, {1,3}},
{{1}, {2,3}}, {{1}, {1,2,3}}, {{2}, {}}, {{2}, {1}}, {{2}, {2}}, {{2}, {3}}, {{2}, {1,2}},{{2}, {1,3}}, {{2}, {2,3}}, {{2}, {1,2,3}}, {{3}, {}}, {{3}, {1}}, {{3}, {2}}, {{3}, {3}},
{{3}, {1,2}}, {{3}, {1,3}}, {{3}, {2,3}}, {{3}, {1,2,3}}, {{1,2}, {}}, {{1,2}, {1}},{{1,2}, {2}}, {{1,2}, {3}}, {{1,2}, {1,2}}, {{1,2}, {1,3}}, {{1,2}, {2,3}}, {{1,2}, {1,2,3}},
{{1,3}, {}}, {{1,3}, {1}}, {{1,3}, {2}}, {{1,3}, {3}}, {{1,3}, {1,2}}, {{1,3}, {1,3}},{{1,3}, {2,3}}, {{1,3}, {1,2,3}}, {{2,3}, {}}, {{2,3}, {1}}, {{2,3}, {2}}, {{2,3}, {3}},
{{2,3}, {1,2}}, {{2,3}, {1,3}}, {{2,3}, {2,3}}, {{2,3}, {1,2,3}}, {{1,2,3}, {}},{{1,2,3}, {1}}, {{1,2,3}, {2}}, {{1,2,3}, {3}}, {{1,2,3}, {1,2}}, {{1,2,3}, {1,3}},
{{1,2,3}, {2,3}}, {{1,2,3}, {1,2,3}}}

```
relconj
```

{{{}, {}}, {{}, {1}}, {{}, {2}}, {{}, {3}}, {{}, {1,2}}, {{}, {1,3}}, {{}, {2,3}},{{}, {1,2,3}}, {{1}, {1}}, {{1}, {1,2}}, {{1}, {1,3}}, {{1}, {1,2,3}}, {{2}, {2}},
{{2}, {1,2}}, {{2}, {2,3}}, {{2}, {1,2,3}}, {{3}, {3}}, {{3}, {1,3}}, {{3}, {2,3}},{{3}, {1,2,3}}, {{1,2}, {1,2}}, {{1,2}, {1,2,3}}, {{1,3}, {1,3}}, {{1,3}, {1,2,3}},
{{2,3}, {2,3}}, {{2,3}, {1,2,3}}, {{1,2,3}, {1,2,3}}}

```
ordenQ[{x, relconj}]
```

True

```
reldiv:=Select[cartesiano[x], Mod[#[[2]], #[[1]]]==0&]

x=Join[Range[-5,-1], Range[1,5]]
```

{−5, −4, −3, −2, −1, 1, 2, 3, 4, 5}

```
ordenQ[{x,reldiv}]
```

False

```
antisimetricaQ[{x,reldiv}]
```

False

```
x=Range[1,10]
```

{1,2,3,4,5,6,7,8,9,10}

```
ordenQ[{x,reldiv}]
```

True

```
reldivest:=Select[cartesiano[x],(Mod[#[[2]],#[[1]]]==0∧#[[1]]≠#[[2]])&]
```

```
ordenQ[{x,reldivest}]
```

False

```
reldivest
```

{{1,2}, {1,3}, {1,4}, {1,5}, {1,6}, {1,7}, {1,8}, {1,9}, {1,10}, {2,4}, {2,6}, {2,8}, {2,10}, {3,6}, {3,9}, {4,8}, {5,10}}

```
relmenorigual:=Select[cartesiano[x], #[[1]]≤ #[[2]]&]
```

```
relmenorigual
```

{{1,1}, {1,2}, {1,3}, {1,4}, {1,5}, {1,6}, {1,7}, {1,8}, {1,9}, {1,10}, {2,2}, {2,3}, {2,4}, {2,5}, {2,6}, {2,7}, {2,8}, {2,9}, {2,10}, {3,3}, {3,4}, {3,5}, {3,6}, {3,7}, {3,8}, {3,9}, {3,10}, {4,4}, {4,5}, {4,6}, {4,7}, {4,8}, {4,9}, {4,10}, {5,5}, {5,6}, {5,7}, {5,8}, {5,9}, {5,10}, {6,6}, {6,7}, {6,8}, {6,9}, {6,10}, {7,7}, {7,8}, {7,9}, {7,10}, {8,8}, {8,9}, {8,10}, {9,9}, {9,10}, {10,10}}

```
ordentotalQ[{x, relmenorigual}]
```

True

```
ordenQ[{x,reldiv}]
```

True

```
ordentotalQ[{x,reldiv}]
```

True

La definición de conjunto ordenado puede hacerse también a partir de la noción de *orden estricto*.

Definición 3. Sea X un conjunto, y $<$ una relación binaria en X. Se dice que $<$ es un orden estricto si se verifican las siguientes propiedades:

Antirreflexiva Para cualquier $x \in X$ se tiene que $x \not< x$.

Transitiva Si $x < y$ e $y < z$ entonces $x < z$.

Es fácil comprobar que si $\leq$ es una relación de orden en un conjunto X, entonces si definimos

$$x < y \text{ si } x \leq y \text{ y } x \neq y$$

se tiene que $<$ es una relación de orden estricto en X.

De la misma forma, si $<$ es una relación de orden estricto en X entonces la relación siguiente:

$$x \leq y \text{ si } x < y \text{ o } x = y$$

es una relación de orden en X.

Vemos entonces que los conceptos de *relación de orden* y *relación de orden estricto* son equivalentes, pues dada una relación de orden tenemos determinada una relación de orden estricto y viceversa. Además, los caminos para pasar de orden a orden estricto, y de orden estricto a orden, son uno el inverso del otro.

```
ordenestricotQ[{_, {es___,{x_,x_}, fs___}}]:=False;
ordenestrictpQ[q_]:=transitivaQ[q];
```

```
ordentotalQ[{x, reldivest}]
```

False

```
ordenestrictoQ[{x, reldivest}]
```

True

A continuación vamos a construir un grafo (dirigido) asociado a una relación de orden. Aún cuando los grafos serán estudiados con posterioridad, la representación de una relación de orden mediante este grafo ayuda a visualizar mejor el orden dado.

Definición 4. El diagrama de Hasse de un conjunto ordenado $(X, \leq)$ es un grafo dirigido cuyos vértices son los elementos de X, y existe un lado de x a y si $x < y$ y no existe z tal que $x < z < y$.

El diagrama de Hasse de un conjunto ordenado está definido para cualquier conjunto ordenado. Sin embargo, en general dicho diagrama no permite recuperar el orden. Por ejemplo, en el caso del conjunto $(\mathbb{R}, \leq)$, dado cualquier $x \in \mathbb{R}$ no existe ningún $y \in \mathbb{R}$ que esté conectado a x por algún lado.

Sin embargo, si el conjunto X es finito, entonces dados $x, y \in X$ se tiene que $x \leq y$ si $x = y$ o existe algún camino que parta de x y termine en y.

Una forma habitual de representar el diagrama de Hasse es dibujar los lados como líneas ascendentes, lo que implica colocar los vértices de forma apropiada.

Ejemplo 5.

1. Consideramos el conjunto ordenado $(\mathscr{P}(\{1,2\}, \subseteq)$. Entonces el diagrama de Hasse sería:



2. Vamos a representar los diagramas de Hasse de los conjuntos ordenados $\mathscr{P}(\{1,2,3\})$ y D(30).





Observa como la estructura de conjunto ordenado es igual en ambos casos.

3. Cualquier grafo dirigido que no contenga caminos cerrados es el diagrama de Hasse de un conjunto ordenado. Así, dado el grafo dirigido



tenemos definido un orden en el conjunto $X = \{a, b, c, d, e, f, g, h, i\}$. Con este orden se tiene, por ejemplo que,

$h \leq e$, pues tenemos un camino $h - f - e$ que empieza en h y termina en e.

$i \leq a$, pues el camino $i - g - d - a$ empieza en i y termina en a.

$i \nleq e$, pues ningún camino empieza en i y termina en e.

```
enmedioQ[{x_,y_}, {es___,{x_,z_}, fs___,{z_,y_},gs___}] := True /;
   Not[MemberQ[{x,y},z]]
enmedioQ[{x_,y_}, {es___,{z_,y_}, fs___,{x_,z_},gs___}] := True /;
   Not[MemberQ[{x,y},z]]
enmedioQ[_, _] := False;

diagramahasse[{x_,r_}]:={x,Select[r,((#[[1]]=!=#[[2]])∧Not[enmedioQ[#,r]])&]} /;
   ordenQ[{x,r}]

x=subconjuntos[{1,2}]
```

{{}, {1}, {2}, {1,2}}

```
diagramahasse[{x,relconj}]
```

{{{}, {1}, {2}, {1,2}}, {{{}, {1}}, {{}, {2}}, {{1}, {1,2}}, {{2}, {1,2}}}}

```
x=subconjuntos[{1,2,3}]
```

{{}, {1}, {2}, {3}, {1,2}, {1,3}, {2,3}, {1,2,3}}

```
d1=diagramahasse[{x,relconj}]
```

{{{}, {1}, {2}, {3}, {1,2}, {1,3}, {2,3}, {1,2,3}},

{{{}, {1}}, {{}, {2}}, {{}, {3}}, {{1}, {1,2}}, {{1}, {1,3}}, {{2}, {1,2}},
{{2}, {2,3}}, {{3}, {1,3}}, {{3}, {2,3}}, {{1,2}, {1,2,3}}, {{1,3}, {1,2,3}},
{{2,3}, {1,2,3}}}}

```
d1/.{{}→1, {1}→2, {2}→3, {3}→4, {1,2}→ 5,{1,3}→6, {2,3}→7, {1,2,3}→8}
```

{{1,2,3,4,5,6,7,8},

{{1,2}, {1,3}, {1,4}, {2,5}, {2,6}, {3,5}, {3,7}, {4,6}, {4,7}, {5,8}, {6,8}, {7,8}}}

```
x=Divisors[30]
```

$\{1,2,3,5,6,10,15,30\}$

```
d2=diagramahasse[{x,reldiv}]
```

$\{\{1,2,3,5,6,10,15,30\},$

$\{\{1,2\}, \{1,3\}, \{1,5\}, \{2,6\}, \{2,10\}, \{3,6\}, \{3,15\}, \{5,10\}, \{5,15\}, \{6,30\}, \{10,30\}, \{15,30\}\}\}$

```
d2/.{5→4, 6→5, 10→6, 15→7, 30→8}
```

$\{\{1,2,3,4,5,6,7,8\},$

$\{\{1,2\}, \{1,3\}, \{1,4\}, \{2,5\}, \{2,6\}, \{3,5\}, \{3,7\}, \{4,6\}, \{4,7\}, \{5,8\}, \{6,8\}, \{7,8\}\}\}$

Definición 6. Sea $(X,\leq)$ un conjunto ordenado.

1. Un elemento $x \in X$ se dice que es maximal, si no existe $y \in X$ tal que $x \leq y$ y $x \neq y$.
2. Un elemento $x \in X$ se dice que es máximo, si para todo $y \in X$ se verifica que $y \leq x$.

De la misma forma se puede definir lo que es un elemento minimal y lo que es un mínimo.

Ejemplo 7. En el último conjunto ordenado del ejemplo anterior se tiene que a y b son elementos maximales, pues no hay ningún elemento que sea mayor que ellos. Sin embargo, el conjunto X no tiene máximo.

El elemento j es un elemento minimal, y además es mínimo.

En el conjunto de los divisores de 30 (ver ejemplo anterior) tenemos que 10 no es un elemento maximal, pues $10 \leq 30$. Sí se tiene que 30 es un elemento maximal, pues no hay ningún elemento que sea mayor que él. También se tiene que 30 es un máximo de ese conjunto.

Nótese, que si un conjunto tiene máximo, entonces este es único. Además, en el caso de que tenga máximo, entonces tiene sólo un elemento maximal, que coincide con el máximo.

Idéntica observación vale para mínimo y elemento minimal.

Denotaremos por $\max(X)$ al máximo del conjunto X, en el caso de que exista, y por $\min(X)$ al mínimo.

En el ejemplo que hemos estudiado anteriormente no existe $\max(X)$, mientras que $\min(X) = j$.

```
maximalQ[x_,{v_,_}]:=False /; Not[MemberQ[v,x]]
maximalQ[x_,{_,{___,{x_,y_},}}]:=False /; (x=!=y)
maximalQ[_,{_,_}]:=True
```

```
x=Divisors[30]
```

{1,2,3,5,6,10,15,30}

```
maximalQ[10, {x,reldiv}]
```

False

```
maximalQ[30, {x,reldiv}]
```

True

```
maximales[{v_,r_}]:=Select[v, maximalQ[#,{v,r}]&]
```

```
maximales[{x,reldiv}]
```

{30}

```
reltrunc:=Complement[reldiv,{{15,30}}]
```

```
maximales[{x,reltrunc}]
```

{15,30}

```
maximo[{x_,r_}]:=With[{ms=maximales[{x,r}]},
     If[Length[ms]==1, ms[[1]], Print["No tiene máximo"]]]
```

```
maximo[{x,reltrunc}]
```

No tiene máximo

```
maximo[{x,reldiv}]
```

30

```
minimalQ[x_,{v_,_}]:=False /; Not[MemberQ[v,x]]
minimalQ[x_,{_,{___,{y_,x_},}}]:=False /; (x=!=y)
minimalQ[_,{_,_}]:=True

minimales[{v_,r_}]:=Select[v, minimalQ[#,{v,r}]&]

minimo[{x_,r_}]:=With[{ms=minimales[{x,r}]},
     If[Length[ms]==1, ms[[1]], Print["No tiene mínimo"]]]

minimalQ[f,{{a,c,d,f,l},  {{l,l}, {a,l}, {a,a}, {a,c}, {a,d}, {a,f}, {c,l},
                            {c,c}, {c,f}, {d,l}, {d,d}, {d,f}, {f,l}, {f,f}}}]
```

False

```
minimales[{x,reldiv}]
```

{1}

```
minimales[{x,reltrunc}]
```

{1}

```
minimo[{x,reldiv}]
```

1

Definición 8. Sea $(X,\leq)$ un conjunto ordenado, e Y un subconjunto de X. Consideramos en Y el orden inducido de X.

1. Un elemento $x\in X$ se dice que es *cota superior* de Y si $x\geq y$ para todo $y\in Y$.
2. Un elemento $x\in X$ se dice que es *supremo* de Y si es el mínimo del conjunto de las cotas superiores de Y.

De la misma forma se define lo que es una cota inferior y un ínfimo.

Ejemplo 9. Si $X = \{a, b, c, d, e, f, g, h, i, j\}$ con el orden dado anteriormente, e $Y = \{c, d, f, g, h\}$ entonces:

El conjunto de las cotas superiores de Y es $\{a\}$.

Puesto que este conjunto tiene mínimo, que es a, entonces a es el supremo de Y.

Los elementos c y d son elementos maximales de Y.

El conjunto de las cotas inferiores es $\{h, j\}$.

De éstas, h es el máximo, luego h es el ínfimo de Y.

h es además el único elemento minimal y el mínimo de Y.

```
cotasuperior[x_,y_,{v_,r_}]:=(Join[{x},y]⊆v)∧(And@@(MemberQ[r,{#,x}]&/@y))
```

```
{x,reldiv}
```

$\{\{1,2,3,5,6,10,15,30\}, \{\{1,1\},\{1,2\},\{1,3\},\{1,5\},\{1,6\},\{1,10\},\{1,15\},\{1,30\},\{2,2\},\{2,6\},\{2,10\},\{2,30\},\{3,3\},\{3,6\},\{3,15\},\{3,30\},\{5,5\},\{5,10\},\{5,15\},\{5,30\},\{6,6\},\{6,30\},\{10,10\},\{10,30\},\{15,15\},\{15,30\},\{30,30\}\}\}$

```
cotasuperior[30, {2,10,15}, {x,reldiv}]
```

True

```
cotasuperior[15, {2,10,15}, {x,reldiv}]
```

False

```
cotassuperiores[y_, {v_,r_}]:=Select[v, cotasuperior[#,y,{v,r}]&]
```

```
cotassuperiores[{2,6,10},{x,reldiv}]
```

{30}

```
cotainferiorQ[x_,y_,{v_,r_}]:=(Join[{x},y]⊆v)∧(And@@(MemberQ[r,{#,x}]&/@y))

cotasinferiores[y_, {v_,r_}]:=Select[v, cotainferiorQ[#,y,{v,r}]&]

cotasinferiores[{2,6,10},{x,reldiv}]
```

{1,2}

```
supremo[y_, {v_,r_}]:=Module[{cs, rtrunc, min},
     cs=cotassuperiores[y, {v,r}];
     rtrunc=Select[r,#⊆cs&]; (*selecciono sólo las relaciones
                              que mueven las cotas superiores*)
     min=minimales[{cs, rtrunc}];
     If[Length[min]==1,  First[min], Print[y, "no tiene supremo"]]
  ]

supremo[{2,6,10}, {x,reldiv}]
```

30

```
infimo[y_, {v_,r_}]:=Module[{ci, rtrunc, max},
     ci=cotasinferiores[y, {v,r}];
     rtrunc=Select[r, #⊆ci&]; (*selecciono sólo las relaciones
                               que mueven las cotas inferiores*)
     max=maximales[{ci, rtrunc}];
     If[Length[max]==1, First[max], Print[y, "no tiene infimo"];Null]
  ]

[{2,6,10}, {x,reldiv}]
```

2

```
infimo[{6,10}, {x,Complement[reldiv, {{1,2}}]}]==Null
```

{6,10} no tiene ínfimo

True

```
cotasinferiores[{6,10}, {x,Complement[reldiv, {{1,2}}]}]
```

{1,2}

Cuando un conjunto tiene supremo éste es único. Podemos entonces hablar de *el supremo de* Y, y lo representaremos mediante $\sup(Y)$.

De la misma forma, denotaremos por $\inf(Y)$ al ínfimo del conjunto Y cuando exista.

Cuando un conjunto tiene máximo, entonces también tiene supremo, y coincide con él. En el último ejemplo vemos como el recíproco no es cierto, pues Y tiene supremo pero no tiene máximo.

Cuando el supremo de un conjunto pertenezca al conjunto, entonces será también el máximo.

Definición 10 (Buen orden)**.** Sea $(X,\leq)$ un conjunto ordenado. Se dice que $\leq$ es un buen orden si todo subconjunto no vacío de X tiene mínimo. En tal caso, se dice que $(X,\leq)$ (o X) es un conjunto bien ordenado.

Observación: Todo conjunto bien ordenado es un conjunto totalmente ordenado, pues dados dos elementos $x, y \in X$ el subconjunto $\{x, y\}$ tiene mínimo. Si $\min(\{x, y\}) = x$ entonces $x \leq y$, mientras que si $\min(\{x, y\}) = y$ entonces $y \leq x$.

El recíproco no es cierto. Busca un ejemplo.

Ejemplo 11. El conjunto de los números naturales, con el orden usual, es un conjunto bien ordenado, como demostramos en el primer tema.

Definición 12. Sean $(X_1,\leq_1)$ y $(X_2,\leq_2)$ dos conjuntos ordenados.

Se define el *orden producto* en $X_1 \times X_2$ como sigue:

$$(x_1, x_2) \preceq (y_1, y_2) \text{ si } x_1 \leq_1 y_1 \text{ y } x_2 \leq_2 y_2.$$

Se define el *orden lexicográfico* en $X_1 \times X_2$ como sigue:

$$(x_1, x_2) \leq_{lex} (y_1, y_2) \overset{\text{def}}{\Longleftrightarrow} \begin{cases} x_1 <_1 y_1 & \text{ó} \\ x_1 = y_1 \text{ y } x_2 \leq_2 y_2. \end{cases}$$

Claramente, si $(x_1, x_2) \preceq (y_1, y_2)$ entonces $(x_1, x_2) \leq_{lex} (y_1, y_2)$.

Proposición 13. *Si $(X_1, \leq_1)$ y $(X_2, \leq_2)$ son dos conjuntos ordenados, entonces $(X_1 \times X_2, \preceq)$ y $(X_1 \times X_2, \leq_{lex})$ son conjuntos ordenados.*

Además, si $\leq_1$ y $\leq_2$ son órdenes totales (resp. buenos órdenes) entonces $\leq_{lex}$ es un orden total (resp. buen orden).

Demostración. La demostración de que el orden producto es una relación de orden es fácil, y se deja como ejercicio. Centrémonos pues en el orden lexicográfico.

Notemos en primer lugar que si $(x_1, x_2) \leq_{lex} (y_1, y_2)$ entonces $x_1 \leq_1 y_1$.

Veamos que la relación es de orden.

Reflexiva: Si $(x_1, x_2) \in X_1 \times X_2$ entonces $(x_1, x_2) \leq_{lex} (x_1, x_2)$, pues se da la segunda opción ($x_1 = x_1$ y $x_2 \leq x_2$).

Simétrica: Supongamos que $(x_1, x_2) \leq_{lex} (y_1, y_2)$ y $(y_1, y_2) \leq_{lex} (x_1, x_2)$. Entonces se tiene que $x_1 \leq_1 y_1$ e $y_1 \leq_1 x_1$, de donde $x_1 = y_1$. Deducimos entonces que $x_2 \leq_2 y_2$ e $y_2 \leq_2 x_2$, lo que implica que $x_2 = y_2$.

Transitiva: Supongamos ahora que $(x_1, x_2) \leq_{lex} (y_1, y_2)$ y $(y_1, y_2) \leq_{lex} (z_1, z_2)$. Pueden darse entonces tres opciones(no excluyentes):

- $x_1 <_1 y_1$, en cuyo caso $x_1 <_1 z_1$, luego $(x_1, x_2) \leq_{lex} (z_1, z_2)$.
- $y_1 <_1 z_1$, en cuyo caso $x_1 <_1 z_1$ y concluimos como en la opción anterior.
- $x_1 = y_1$ e $y_1 = z_1$. En tal caso, $x_2 \leq_2 y_2$ e $y_2 \leq_2 z_2$, de donde $x_1 = z_1$ y $x_2 \leq_2 z_2$, es decir, $(x_1, x_2) \leq_{lex} (z_1, z_2)$.

Supongamos ahora que $\leq_1$ y $\leq_2$ son órdenes totales. Sean $(x_1, x_2), (y_1, y_2) \in X_1 \times X_2$. Aquí pueden darse tres opciones (mutuamente excluyentes):

- $x_1 <_1 y_1$. En tal caso $(x_1, x_2) \leq_{lex} (y_1, y_2)$.
- $y_1 <_1 x_1$. En este caso $(y_1, y_2) \leq_{lex} (x_1, x_2)$.
- $x_1 = y_1$. Entonces dependiendo de que $x_2 \leq_2 y_2$ o $y_2 \leq_2 x_2$ se tendrá que $(x_1, x_2) \leq_{lex} (y_1, y_2)$ o que $(y_1, y_2) \leq_{lex} (x_1, x_2)$.

Por último, supongamos que $\leq_1$ y $\leq_2$ son buenos órdenes, y sea $Y \subseteq X_1 \times X_2$ un subconjunto no vacío.
Nos quedamos con el conjunto de todas las primeras coordenadas de los elementos de A, es decir, tomamos

$$Y_1 = \{x_1 \in X_1 : (x_1, x_2) \in A \text{ para algún } x_2 \in X_2\}.$$

Sea $a = \text{mín}(Y_1)$. Tomamos entonces $Y_2 = \{x_2 \in X_2 : (a, x_2) \in A\}$. Como $Y_2 \neq \emptyset$, tiene mínimo. Sea éste b. Entonces $(a, b) = \text{mín}(A)$. □

Observación: Si tenemos n conjuntos ordenados $X_1, X_2, \ldots, X_n$, podemos definir recursivamente el orden producto y el orden lexicográfico en $X_1 \times X_2 \times \cdots \times X_n$.

Supuesto definido el orden producto $\preceq$ en $X_1 \times \cdots \times X_{n-1}$ se define en $X_1 \times \cdots \times X_n$:

$$(x_1, \ldots x_{n-1}, x_n) \preceq (y_1, \ldots, y_{n-1}, y_n) \; si \; (x_1, \ldots, x_{n-1}) \preceq (y_1, \ldots, y_{n-1}) \text{ y } x_n \leq y_n,$$

es decir, definimos el orden producto en $(X_1 \times \cdots \times X_{n-1}) \times X_n$.

Supuesto definido el orden lexicográfico $\leq_{lex}$ en $X_1 \times \cdots \times X_{n-1}$ se define en $X_1 \times \cdots \times X_n$:

$$(x_1, \ldots x_{n-1}, x_n) \leq_{lex} (y_1, \ldots, y_{n-1}, y_n) \overset{\text{def}}{\Longleftrightarrow} \begin{cases} (x_1, \ldots, x_{n-1}) <_{lex} (y_1, \ldots, y_{n-1}) & \text{ó} \\ (x_1, \ldots, x_{n-1}) = (y_1, \ldots, y_{n-1}) \; y \; x_n \leq y_n. \end{cases}$$

Sea el conjunto

$$\mathscr{A} = \{\ , a, b, c, d, e, f, g, h, i, j, l, l, m, n, ñ, o, p, q, r, s, t, u, v, w, x, y, z\},$$

es decir, las 27 letras del alfabeto junto con el espacio en blanco.
Claramente, $\mathscr{A}$ tiene un orden total de todos conocido.
Supongamos que n es el número de letras de la palabra más larga de la lengua española. Entonces, cada palabra puede representarse como un elemento de $\mathscr{A}^n$ (poniendo tantos espacios al final como sea necesario).
Cuando ordenamos las palabras, tal y como vienen en un diccionario, nos fijamos en la primera letra, y es la que nos da el orden. Cuando ésta coincide, pasamos a la segunda, y es ésta entonces la que nos da el orden. De coincidir también, nos fijamos en la tercera, y así sucesivamente. Es decir, las palabras de la lengua están ordenadas siguiendo el orden lexicográfico.

Ejemplo 14.

1. Consideramos en $\mathbb{N} \times \mathbb{N}$ los órdenes producto ($\preceq$) y lexicográfico $\leq_{lex}$ deducidos a partir del orden usual en $\mathbb{N}$. Entonces:

   Los elementos $(0, n), (1, n-1), \ldots, (n-1, 1), (n, 0)$ están ordenados según el orden lexicográfico, mientras que con el orden producto ninguna pareja de ellos es comparable.

   Se puede ver entonces que la propiedad de ser orden total o buen orden no se mantiene al tomar el orden producto.

   Si $X = \{(0, n), (1, n-1), \ldots, (n-1, 1), (n, 0)\}$ entonces:

- El conjunto de cotas inferiores con respecto al orden lexicográfico es $\{(0,0),(0,1),\ldots,(0,n)\}$, mientras que con respecto al orden producto tiene una única cota inferior, que es $(0,0)$.
- El ínfimo, respecto al orden lexicográfico es $(0,n)$, que es también el mínimo. Con respecto al orden producto es $(0,0)$, y no tiene mínimo.
- Con respecto al orden lexicográfico tiene un elemento minimal, que es $(0,n)$ y un elemento maximal, que es $(n,0)$. Con respecto al orden producto, todos los elementos son maximales y minimales.

```
menorlexQ[{x_,xs___},{x_,ys___}]:=menorlexQ[{xs},{ys}];
menorlexQ[{},_]:=True;
menorlexQ[{x_,___},{y_,___}]:=x<y;
```

```
menorlexQ[{1,2,3}, {1,1,20}]
```

True

```
menorlexQ[{1,2,3}, {3,2,1}]
```

False

```
x=Map[{#, 3-#}&, Range[0,3]]
```

{{0,3}, {1,2}, {2,1}, {3,0}}

```
xx=cartesiano[Range[0,3]]
```

{{0,0}, {0,1}, {0,2}, {0,3}, {1,0}, {1,1}, {1,2}, {1,3},
{2,0}, {2,1}, {2,2}, {2,3}, {3,0}, {3,1}, {3,2}, {3,3}}

```
rellex=Select[cartesiano[xx], menorlexQ[#[[1]], #[[2]]]&];
```

```
cotasinferiores[x,{xx,rellex}]
```

{{0,0}, {0,1}, {0,2}, {0,3}}

```
infimo[x,{xx,rellex}]
```

{0,3}

```
minimo[{x,Select[rellex,#⊆x&]}]
```

{0,3}

En los ejemplos de órdenes que hemos introducido en Mathematica, ha sido necesario especificar todas las parejas (x, y) tales que $x \leq y$. Gran parte de esta información es redundante. De hecho, hemos visto como podemos determinar relaciones de orden dando únicamente el diagrama de Hasse. En él se especifican únicamente algunas parejas de elementos relacionados, y a partir de ellas se deducen todas las demás. Esto se consigue haciendo uso de las propiedades que definen un conjunto ordenado. En particular, la propiedad reflexiva y la transitiva.

Así, en un diagrama de Hasse, no se especifica una relación del tipo $a \leq a$, pues esa está implícita en la definición de relación de orden. También, si tenemos que $a \leq b$ y $b \leq c$, no es necesario destacar la relación $a \leq c$, pues esta se deduce de las dos anteriores.

Lo que tratamos a continuación es, dada una relación binaria en un conjunto, construir la relación de orden que se puede deducir de ella. Para esto, añadiremos todas las relaciones de la forma $a \leq a$ (a esto lo denominaremos *cierre reflexivo*, y si tenemos $a \leq b$ y $b \leq c$, añadiremos $a \leq c$ (*cierre transitivo*).

```
cierrereflexivo[{v_,r_}]:={v,Union[r,Map[{#,#}&,v]]}
```

```
cierretransitivo[{v_,r:{as___,{x_,y_},bs___,{y_,z_},cs___}}]:=
   cierretransitivo[{v,Union[{as,{x,y},bs,{y,z},{x,z},cs}]}] /;
      Not[MemberQ[r,{x,z}]]∧Length[Union[{x,y,z}]]==3
cierretransitivo[{v_,r:{as___,{y_,z_},bs___,{x_,y_},cs___}}]:=
   cierretransitivo[{v,Union[{as,{x,y},bs,{y,z},{x,z},cs}]}] /;
      Not[MemberQ[r,{x,z}]]∧Length[Union[{x,y,z}]]==3
cierretransitivo[{v_,r_}]:={v,r}
```

```
cierretransitivo[{{1,2,3},  {{1,2}, {2,3}}}]
```

{{1,2,3}, {{1,2}, {1,3}, {2,3}}}

```
clausura[x_]:=cierrereflexivo[cierretransitivo[x]]
```

```
x=Divisors[30]
```

{1,2,3,5,6,10,15,30}

```
diagramahasse[{x,reldiv}]
```

{{1,2,3,5,6,10,15,30}, {{1,2}, {1,3}, {1,5}, {2,6}, {2,10}, {3,6}, {3,15}, {5,10}, {5,15}, {6,30}, {10,30}, {15,30}}

```
clausura[diagramahasse[{x,reldiv}]]
```

{{1,2,3,5,6,10,15,30}, {{1,1}, {1,2}, {1,3}, {1,5},{1,6}, {1,10}, {1,15}, {1,30}, {2,2}, {2,6}, {2,10}, {2,30}, {3,3}, {3,6}, {3,15}, {3,30}, {5,5}, {5,10}, {5,15}, {5,30}, {6,6}, {6,30}, {10,10}, {10,30}, {15,15}, {15,30}, {30,30}}

```
{x,reldiv}
```

{{1,2,3,5,6,10,15,30}, {{1,1}, {1,2}, {1,3}, {1,5},{1,6}, {1,10}, {1,15}, {1,30}, {2,2}, {2,6}, {2,10}, {2,30}, {3,3}, {3,6}, {3,15}, {3,30}, {5,5}, {5,10}, {5,15}, {5,30}, {6,6}, {6,30}, {10,10}, {10,30}, {15,15}, {15,30}, {30,30}}

```
%==&&
```

True
